



プログラミング演習2 クラスと継承

中村, 青山, 小林, 橋本

クラスの定義



- 財布クラスを作るにはどうする？
 - 内部的に持つ情報は？ 必要なメソッドは？
- 人情報を管理するクラスを作るにはどうする？
 - 姓, 名, 年齢
- ペイントツールで1つずつのストロークを管理するには？
- 文字列を扱うにはどうするか？



- String 型は, 文字列を扱うためのクラス
 - 「中村聡史」「明治大学 総合数理学部」
- 文字列を扱う際にはこういった機能が必要？
 - 文字列の長さを取得する
 - 文字列にある文字が含まれているかを調べる
 - 文字列を部分的に置き換える
 - 文字列が一致しているか調べる
 - n文字目の文字を取得する
 - などなど

Stringクラスのメソッド



- `charAt(num);` `num`文字目の文字を返す(0から始まる)
- `indexOf(文字列);` 入力された文字列が何文字目か?
- `length();` 入力された文字の文字数を返す
- `substring(x);` `x`文字目から最後までを出力
- `substring(x, y);` `x`文字目から`y-1`文字目までを出力
- `toLowerCase();` 全てを小文字に変換する
- `toUpperCase();` 全てを大文字に変換する
- `replace(文字列A, 文字列B);`
 - 文字列Aを文字列Bに変更する
- `split(文字列);` 文字列を分割

<http://processing.org/reference/String.html>

<http://docs.oracle.com/javase/6/docs/api/java/lang/String.html>

```
String str = "Department of Frontier Media Science (FMS), IMS, Meiji University";

println( str.length() );
println( str.charAt( 11 ) );
println( str.indexOf("F") );
println( str.indexOf("S") );
println( str.indexOf("Meiji") );
println( str.substring( str.indexOf("Meiji") ) );
println( str.toLowerCase() );
println( str.toUpperCase() );
```



```
65
o
14
29
49
Meiji University
department of frontier media science (fms), ims, meiji university
DEPARTMENT OF FRONTIER MEDIA SCIENCE (FMS), IMS, MEIJI UNIVERSITY
```

```
String str = "Department of Frontier Media Science (FMS), IMS, Meiji University";

String [] ret = str.split( " " );
println( ret.length ) ;
int i=0;
while( i<ret.length ){
    println( ret[i] );
    i++;
}
```



```
9
Department
of
Frontier
Media
Science
(FMS),
IMS,
Meiji
University
```



- 下記の結果は？

```
println( str.indexOf("Media") );  
println( str.toLowerCase().indexOf("media") );  
println( str.substring(  
    str.indexOf("Meiji") ).length() );
```

- Frontier Media Science を出力するには？

```
str.substring(           ,           );
```



- 猫, 犬, 猿, 象, 熊を定義

- それぞれの座標は意識したくない

- `cat.x`, `cat.y`, `dog.x`, `dog.y`, `monkey.x`, `monkey.y`, ...
- 内部で適当に処理してもらう

- 描画はシンプルにしたい

- `cat.draw()`, `dog.draw()`, `monkey.draw()`, `elephant.draw()`, ...

- 移動もシンプルにしたい

- `cat.move()`, `dog.move()`, `monkey.move()`, `elephant.move()`, ...

- 睡眠も任せてしまう

- `cat.sleep()`, `dog.sleep()`, `monkey.sleep()`, `elephant.sleep()`, ...


いいんだけど、重複があるのでは？

×が動き回るクラスを作ろう



Ballクラスを利用して、Crossクラスを作ろう！

- Ballクラスと、Crossクラスの違いは、表示される図形が「○」か「×」かなだけ！！
- Crossクラスのタブを作成し、Ballクラスをコピー！
- BallをCrossに書き換える！
- 表示だけを変更したいので、displayの中身を変更する！
- メインのプログラムで Cross を使っていこう！



Ballを改良しCrossを作る



```
class Cross{
  int x;
  int y;
  int speedX;
  int speedY;

  Cross(){
    init();
  }
  void init(){
    x = (int)random(width);
    y = (int)random(height);
    speedX = (int)random(5);
    speedY = (int)random(5);
  }
  void display(){
    line( x-15, y-15, x+15, y+15 );
    line( x+15, y-15, x-15, y+15 );
  }
}
```

```
void move(){
  x = x + speedX;
  y = y + speedY;
  if ( x+15 > width ) {
    x = width - 15;
    speedX = -speedX;
  }
  if( x - 15 < 0 ){
    x = 15;
    speedX = -speedX;
  }
  if( y + 15 > height ){
    y = height - 15;
    speedY = -speedY;
  }
  if( y - 15 < 0 ){
    y = 15;
    speedY = -speedY;
  }
}
}
```

Crossクラスを使うと



```
Cross fukuchi;  
Cross nakamura;  
Cross hashimoto;  
void setup() {  
    size( 400, 300 );  
    fukuchi = new Cross();  
    nakamura = new Cross();  
    hashimoto = new Cross();  
}  
  
void draw() {  
    background(255);  
    fukuchi.move();  
    nakamura.move();  
    hashimoto.move();  
    fukuchi.display();  
    nakamura.display();  
    hashimoto.display();  
}
```



- Ballクラスと, Crossクラスはほとんど一緒
- 違いはクラス名とコンストラクタ, そしてdisplayのインスタンスメソッドのみ

	Ball	Cross	
インスタンス変数	x, y speedX, speedY	x, y speedX, speedY	一致
クラス名	Ball	Cross	違う
コンストラクタ	Ball()	Cross()	内部は一致
void init()			一致
void move()			一致
void display()	fill(0, 0, 255); ellipse(x, y, 30, 30);	line(x-15, y-15, x+15, y+15); line(x+15, y-15, x-15, y+15);	違う

無駄じゃね？

前回の宿題



- Ball クラスを改良し，× が動き回るCrossクラスと△が動き回るTriangleクラスを作成せよ。
- またこれを利用して，5個の○と，4個の×と，3個の△が動き回るプログラムを作成せよ
 - ただし，その速度は目視可能なものとせよ
 - 1個程度動いていないものがあったてもよい

ほとんど重複している！
違うのはdisplay()だけ！
もっと簡単にできないの？

どう無駄をなくす？



- そもそもBallクラスというのがダメなのでは？
 - Objectクラスという名前にして、objectTypeなどの変数を用意し、displayの時に切り替えては？

```
class Object{
  int x;
  int y;
  int speedX;
  int speedY;
  int objectType;

  void display(){
    if( objectType == 0 ){
      ellipse( x, y, 30, 30 );
    } else if( objectType == 1 ){
      rect( x, y, 30, 30 );
    }
  }
}
```

これも一つの方法
ただ複雑なものだと厳しい

そこで継承！



- 大辞林 第三版

1. 先の人の身分・権利・義務・財産などを受け継ぐこと。「王位を一する」
2. インヘリタンス → (オブジェクト指向プログラミングにおいて、クラス間でデータの共有を行う機構。新しく定義するクラスを既存のクラスの下位クラスとして記述し、上位クラスより属性やメソッドを引き継ぐ仕組みをいう。上位クラスに対する差分のみを記述するだけで新しいクラスを定義することが可能となる。)

一緒に部分をまとめた
スーパークラス(親クラス)を作る

スーパーなObjectクラス



```
class Object{
  int x;
  int y;
  int speedX;
  int speedY;

  Object(){
    init();
  }
  void init(){
    x = (int)random(width);
    y = (int)random(height);
    speedX = (int)random(5);
    speedY = (int)random(5);
  }
}
```

**display() は内容が違うので
削除してしまう**

```
void move(){
  x = x + speedX;
  y = y + speedY;
  if ( x+15 > width ) {
    x = width - 15;
    speedX = -speedX;
  }
  if( x - 15 < 0 ){
    x = 15;
    speedX = -speedX;
  }
  if( y + 15 > height ){
    y = height - 15;
    speedY = -speedY;
  }
  if( y - 15 < 0 ){
    y = 15;
    speedY = -speedY;
  }
}
```


一緒に部分をまとめる



- Ball は Object の変数 (x, y, speedX, speedY) や機能 (移動や初期化) をもち, 独自の表示に関する機能 (メソッド) をもつクラス
- Cross は Object の変数 (x, y, speedX, speedY) や機能 (移動や初期化) をもち, 独自の表示に関する機能 (メソッド) をもつクラス
- インスタンス変数や, インスタンスメソッドを引き継ぐことを**継承**と呼ぶ!

Objectクラスを使うと



```
class Ball extends Object {  
    void display(){  
        fill( 255, 0, 0 );  
        ellipse( x, y, 30, 30 );  
    }  
}  
  
class Cross extends Object {  
    void display(){  
        line( x-15, y-15, x+15, y+15 );  
        line( x+15, y-15, x-15, y+15 );  
    }  
}
```



BallクラスとCrossクラスが劇的に短く！

```
Ball miyashita;  
Ball komatsu;  
Ball kikuchi;  
Cross fukuchi;  
Cross nakamura;  
Cross hashimoto;  
  
void setup() {  
    size( 400, 300 );  
    fill( 255, 0, 0 );  
    miyashita = new Ball();  
    komatsu = new Ball();  
    kikuchi = new Ball();  
    fukuchi = new Cross();  
    nakamura = new Cross();  
    hashimoto = new Cross();  
}  
:
```

なぜ動くの？




- 継承すると、親の力をすべて引き継ぐ！
- 継承の方法は extends とやるだけ！

```
class クラス名 extends 親クラス名 {  
  
}
```

- 継承により親の能力，値はすべて引き継ぎます



- Object クラスを継承して「三角形」を描画するクラスを作るには？ (Triangle)
- Object クラスを継承して「□」を描画するクラスを作るには？ (Sqaure)



Object

x, y
speedX
speedY

init()

move()

Object()

Triangle

display()

三角形と×のクラス



- 動く三角形のクラスと動く×のクラス

```
class Triangle extends Object {
  void display(){
    fill( 0, 255, 0 );
    triangle( x, y-15, x-15, y+15, x+15, y+15 );
  }
}

class Square extends Object {
  void display(){
    fill( 0, 0, 255 );
    rect( x-15, y-15, 30, 30 );
  }
}
```

つまり先週の課題は



- Ball クラスを改良し, × が動き回るCrossクラスと △が動き回るTriangleクラスを作成せよ.
- またこれを利用して, 5個の○と, 4個の×と, 3個の△が動き回るプログラムを作成せよ
 - ただし, その速度は目視可能なものとせよ
 - 1個程度動いていないものがあったてもよい

Objectクラスを継承して
Cross, Triangleクラスを作り,
displayだけ独立させる!

三角形は跳ね返らないように



(Q3) 先述のObjectを継承したTriangleクラスを改良し，三角形は跳ね返らず右端→左端，左端→右端，上端→下端，下端→上端と移動するようにせよ

• 考え方

- Object の move メソッドを Triangle クラス内でオーバーライドして，三角形専用のメソッドを作成する




```
class Triangle extends Object {  
    void display(){  
        fill( 0, 255, 0 );  
        triangle( x, y-15, x-15, y+15, x+15, y+15 );  
    }  
    void move(){  
        x = x + speedX;  
        y = y + speedY;  
        if( x > width ){  
            x = x - width;  
        } else if( x < 0 ){  
            x = width + x;  
        }  
        if( y > height ){  
            y = y - height;  
        }  
        if( x < 0 ){  
            y = width + y;  
        }  
    }  
}
```

move メソッドをオーバーライドして
親の move メソッドが呼ばれないよ
うにする

move をオーバーライド



- Triangle の move で, Object の move を上書きしてしまう!



全体の挙動を変更



(Q4) Objectクラスのmoveメソッドを変更し，上端及び下端は跳ね返るが，左端と右端では逆側から現れるようにせよ

- 考え方

- Object の move メソッドのみ変更
- y 座標の条件で跳ね返り
- x 座標の条件で逆側から現れるようにする

Object クラスのみ変更



```
class Object{
    int x;
    int y;
    int speedX;
    int speedY;

    Object(){
        init();
    }
    void init(){
        x = (int)random(width);
        y = (int)random(height);
        speedX = (int)random(5);
        speedY = (int)random(5);
    }
}
```

```
void move(){
    x = x + speedX;
    y = y + speedY;
    if( x > width ){
        x = x - width;
    } else if( x < 0 ){
        x = width + x;
    }

    if( y + 15 > height ){
        y = height - 15;
        speedY = -speedY;
    } else if( y - 15 < 0 ){
        y = 15;
        speedY = -speedY;
    }
}
}
```



(Q5) 400x300のウィンドウ内で, 任意の場所 x, y から任意の速度で移動する3つの赤色の円を描画し, 右端・左端・上端・下端に来ると跳ね返るようになる. また, 赤色の円には円の中心から30の距離があるところに1つの衛星があり, 10度ずつ円の周りを回転するようにせよ

• 考え方

– 円の中心 (x, y) から衛星の方向の角度 $(0 \sim 360$ 度)を θ とすると, 衛星の座標は

$(x+30*\cos(\text{radians}(\theta)), y+30*\sin(\text{radians}(\theta)))$

惑星と衛星の様なオブジェクト



- Objectクラスを継承して, 変数を追加する


```
class PlanetSatellite extends Object {
  int theta;
  void display(){
    fill( 255, 0, 0 );
    ellipse( x, y, 30, 30 );
    theta = theta + 10;
    int rx = (int)(x+30*sin(radians(theta)));
    int ry = (int)(y+30*cos(radians(theta)));
    ellipse( rx, ry, 10, 10 );
  }
}
```

惑星と衛星の様なオブジェクト

明治大学総合数理学部
先端メディアサイエンス学科
中村研究室




```
PlanetSatellite miyashita;  
PlanetSatellite komatsu;  
PlanetSatellite kikuchi;  
void setup() {  
  size( 400, 300 );  
  miyashita = new PlanetSatellite();  
  komatsu = new PlanetSatellite();  
  kikuchi = new PlanetSatellite();  
}  
  
void draw() {  
  background(255);  
  miyashita.move();  
  komatsu.move();  
  kikuchi.move();  
  miyashita.display();  
  komatsu.display();  
  kikuchi.display();  
}
```



継承すると...



- Object を PlanetSatellite として継承し, theta という変数と, display() というメソッドを追加




惑星と衛星の様なオブジェクト



(Q5) Q4を改良し, 衛星の開始角を0~360度の任意の場所にしたい. どうするか?

- 考え方

- void init() というインスタンスメソッドを追加し, 改良したらOK?



やってみる



```
class PlanetSatellite extends Object {  
    int theta;  
    void init(){  
        theta = (int)random(360);  
    }  
    void display(){  
        fill( 255, 0, 0 );  
        ellipse( x, y, 30, 30 );  
        theta = theta + 10;  
        int rx = (int)(x+30*sin(radians(theta)));  
        int ry = (int)(y+30*cos(radians(theta)));  
        ellipse( rx, ry, 10, 10 );  
    }  
}
```


init() はコンストラクタで呼ばれる
Objectクラス参照

画面の左上から動かないのはなぜ？

上書きしてしまったから



- Object の `init()` を, PlanetSatellite の `init()` でオーバーライドしている (Objectの`init()`が呼び出されない)



やってみる



```
class PlanetSatellite extends Object {
  int theta;
  void init(){
    theta = (int)random(360);
    x = (int)random(width);
    y = (int)random(height);
    speedX = (int)random(5);
    speedY = (int)random(5);
  }
  void display(){
    fill( 255, 0, 0 );
    ellipse( x, y, 30, 30 );
    theta = theta + 10;
    int rx = (int)(x+30*sin(radians(theta)));
    int ry = (int)(y+30*cos(radians(theta)));
    ellipse( rx, ry, 10, 10 );
  }
}
```

Object の init() にあるのを
そのままコピペする

動くけど、なんだか
無駄が増えている...

Objectのinit()も使いたい



- クラスの中で, super と書くと, 継承元の親を呼び出すことができる!

```
class PlanetSatellite extends Object {
  int theta;
  void init(){
    theta = (int)random(360);
    super.init();
  }
  void display(){
    fill( 255, 0, 0 );
    ellipse( x, y, 30, 30 );
    theta = theta + 10;
    int rx = (int)(x+30*sin(radians(theta)));
    int ry = (int)(y+30*cos(radians(theta)));
    ellipse( rx, ry, 10, 10 );
  }
}
```

super . メソッド名

課題3-1: Planet4



- Q4の PlanetSatellite クラスを継承し，衛星の数を4つにする PlanetSatellite4 クラスを作成せよ．また，そのクラスを用いて8個の惑星が動きまわるようにせよ
 - 考え方
 - インスタンス変数を追加する
 - init メソッドと， display メソッドをオーバーライドする

課題3-2: boundAll3



- Objectクラスを継承し, BallクラスとCrossクラス, Triangleクラスを作成せよ
- また, 50個の赤色丸と50個の×と50個の△を動かすようにせよ

課題3-3 DrawChara



- 配布するObjectCharaクラスを継承し、ObjectCharaクラス内のcenterX, centerYを利用してその位置に自身のキャラクタを描画するようにせよ。なお、継承したクラスではdisplay1とdisplay2をオーバーライドせよ
- なおメインプログラムは、キャラクタが動き回り、マウスのボタンが押されていない間は display1 で、押されている間は display2 で描画されるものである
- クラスの名前は名前と番号が含まれるようにし、そのクラスのみ別ファイルとして作成せよ
 - (例) 1-2-33 中村の場合、クラス名を「ExNakamura33」とする
- 来週配布して利用します



- 100秒タイマークラス (TimerBase) を利用して残り時間を表示してみよう
 - ただし、タイマーはクリックの度にカウントダウンを開始し、0秒になったら停止するようにせよ
 - TimerBaseは講義資料サイトからダウンロードしよう
 - また、100秒タイマークラスを継承し2つのタイマーを作成せよ。1つはプログレスバーのスタイルで残り時間を表示するもの、もう1つはなにか適当なものを考えて作成せよ。